Rocket software

Rocket Uniface Library 10.4

Example: Finding a Matching Struct

The following example ProcScript module returns the first Struct node with a specified value in a Struct or collection of Structs. It demonstrates the use of iteration and recursion, access operators, and Struct functiions.

1. Define Struct parameters and variables. This module takes two IN parameters, a Struct or collection of Structs, and a value to search for.
2. Get the total number of Structs using the `$collsize` function.
3. If it is a collection of structs, for each Struct, recursively call this ProcScript module until a match is found.
4. If it is a single Struct, get the number of members using the `$membercount` function.
5. For each member, check whether it is a Struct leaf using the `$isLeaf` function, and evaluate whether the value matches.
6. If it is a Struct node (that is, not a leaf), recursively call this ProcScript module until a match is found.
7. If no match is found, return an empty string.

```
function findFirstMatch
 returns struct
 params   [1]
   struct pStruct: in
   string pMatchValue: in
 endparams
 variables
  struct vNextChild, vMatch
  numeric I, vTotalStructs, vTotalMembers
 endvariables

  vTotalStructs = pStruct->$collsize [2]
  if (vTotalStructs > 1) [3]
    I = 1
    while (I <= vTotalStructs)
      vMatch = findFirstMatch(pStruct{I}, pMatchValue)
      if (vMatch != "")
        return vMatch
      endif
      I = I + 1
    endwhile
  else
    vTotalMembers = pStruct->$membercount [4]
    I = 1
    while (I <= vTotalMembers)
      vNextChild = pStruct->*{I}
      if (vNextChild->$isLeaf) [5]
        if (vNextChild == pMatchValue)
          return vNextChild
        endif
      else
        vMatch = findFirstMatch(vNextChild, pMatchValue) [6]
        if (vMatch != "")
          return vMatch
        endif
      endif
      I = I + 1
```

```
    endwhile
  endif
  return "" [7]
end ;- function findFirstMatch
```